```
> restart; infolevel[dsolve]:=2;
```

$$infolevel_{dsolve} := 2 \quad (1)$$

Empecemos por un principio pediátrico, una ecuación lineal de primer orden inhomogénea

```
> EcDif1 := diff(y(x),x)= exp(-x) -2*y(x);
```

$$EcDif1 := \frac{d}{dx} y(x) = e^{-x} - 2\, y(x) \quad (2)$$

cuya solución es

```
> SolEcDif1 := dsolve(EcDif1,y(x));
Methods for first order ODEs:
--- Trying classification methods ---
trying a quadrature
trying 1st order linear
<- 1st order linear successful
```

$$SolEcDif1 := y(x) = \left(e^{x} + \_C1\right) e^{-2\,x} \quad (3)$$

nótese el efecto de la instrucción infolevel

```
> #?infolevel;
> assign(SolEcDif1);
```

Para graficarla debemos especificar la solución particular, vale decir el valor de las constantes que surgen de las condiciones iniciales. En este caso lo hacemos de manera arbitraria

```
> GrafSolEcDif1part := {seq(subs(_C1 = m, y(x)), m=-4..4)};
```

$$GrafSolEcDif1part := \left\{\left(e^{x} - 4\right) e^{-2\,x}, \left(e^{x} - 3\right) e^{-2\,x}, \left(e^{x} - 2\right) e^{-2\,x}, \left(e^{x} - 1\right) e^{-2\,x}, \left(e^{x} + 1\right) e^{-2\,x}, \left(e^{x} + 2\right) e^{-2\,x}, \left(e^{x} + 3\right) e^{-2\,x}, \left(e^{x} + 4\right) e^{-2\,x}, e^{-2\,x} e^{x}\right\} \quad (4)$$

y graficamos

```
> plot(GrafSolEcDif1part, x=-1..5, -5..5);
```



Pudimos haber dado valores iniciales

```
> CondIni := f(0)=-3;
```

$$CondIni := f(0) = -3 \quad (5)$$

```
> EcDif12 := diff(f(x),x)= exp(-x) -2*f(x);
```

$$EcDif12 := \frac{d}{dx} f(x) = e^{-x} - 2 f(x) \quad (6)$$

```
> SolEcDif12:=dsolve({EcDif12,CondIni},f(x));
Methods for first order ODEs:
--- Trying classification methods ---
trying a quadrature
trying 1st order linear
<- 1st order linear successful
```

$$SolEcDif12 := f(x) = \left(e^{x} - 4\right) e^{-2x} \quad (7)$$

```
> plot(rhs(SolEcDif12), x=-1..5, -5..5);
```



También pudimos haber utilizado el método de las isoclinas

```
> with(DEtools):
> DEplot(EcDif12,f(x),x=-1..5, f= -5..5);
```




```
> DEplot(EcDif12,f(x),x=-1..5,[[0,-3],[0,-2], [0,3],[0,2]],f=-5..5);
dsolve/numeric: entering dsolve/numeric
DEtools/convertsys: converted to first-order system Y'(x) = f(x,Y
(x))    namely (with Y' represented by YP)
```

$$\left[YP_1 = e^{-x} - 2\,Y_1\right]$$

```
DEtools/convertsys: correspondence between Y[i] names and original
functions:
```

$$\left[Y_1 = f(x)\right]$$

```
dsolve/numeric: entering dsolve/numeric
DEtools/convertsys: converted to first-order system Y'(x) = f(x,Y
(x))    namely (with Y' represented by YP)
```

$$\left[YP_1 = e^{-x} - 2\,Y_1\right]$$

```
DEtools/convertsys: correspondence between Y[i] names and original
functions:
```

$$\left[Y_1 = f(x)\right]$$

```
dsolve/numeric: entering dsolve/numeric
DEtools/convertsys: converted to first-order system Y'(x) = f(x,Y
(x))    namely (with Y' represented by YP)
```

$$\left[YP_1 = e^{-x} - 2\,Y_1\right]$$

```
DEtools/convertsys: correspondence between Y[i] names and original
functions:
```

$$\left[Y_1 = f(x)\right]$$

```
dsolve/numeric: entering dsolve/numeric
DEtools/convertsys: converted to first-order system Y'(x) = f(x,Y
(x))    namely (with Y' represented by YP)
```

$$\left[YP_1 = e^{-x} - 2\,Y_1\right]$$

```
DEtools/convertsys: correspondence between Y[i] names and original
functions:
```

$$\left[Y_1 = f(x)\right]$$



Qué hubiera pasado si hubiéramos seleccionado una ecuación diferencial no lineal

```
> EcDif2 := diff(g(x),x) -(2*sqrt(g(x))-g(x))/x = 0;
```

$$EcDif2 := \frac{d}{dx}\,g(x) - \frac{2\sqrt{g(x)} - g(x)}{x} = 0 \qquad \textbf{(8)}$$

Es una ecuación separable y de forma analítica encontramos la familia de soluciones

```
> SolEcDif2 := dsolve(EcDif2,g(x));
```

```
Methods for first order ODEs:
--- Trying classification methods ---
trying a quadrature
trying 1st order linear
trying Bernoulli
<- Bernoulli successful
```

$$SolEcDif2 := \sqrt{g(x)} - 2 - \frac{\_C1}{\sqrt{x}} = 0 \tag{9}$$

y resulta que es una ecuación de Bernoulli,

Bernoulli Differential Equation

$$\frac{dy}{dx} + p(x)\,y = q(x)\,y^n.$$

Let $v \equiv y^{1-n}$ for $n \neq 1$. Then

$$\frac{dv}{dx} = (1-n)\,y^{-n}\,\frac{dy}{dx}.$$

Rewriting (1) gives

$$\begin{aligned} y^{-n}\,\frac{dy}{dx} &= q(x) - p(x)\,y^{1-n} \\ &= q(x) - v\,p(x). \end{aligned}$$

Plugging (4) into (3),

$$\frac{dv}{dx} = (1-n)\,[q(x) - v\,p(x)].$$

Now, this is a linear first-order ordinary differential equation of the form

$$\frac{dv}{dx} + v\,P(x) = Q(x),$$

where $P(x) \equiv (1-n)\,p(x)$ and $Q(x) \equiv (1-n)\,q(x)$. It can therefore be solved analytically using an integrating factor

$$\begin{aligned} v &= \frac{\int e^{\int P(x)\,dx}\,Q(x)\,dx + C}{e^{\int P(x)\,dx}} \\ &= \frac{(1-n)\int e^{(1-n)\int p(x)\,dx}\,q(x)\,dx + C}{e^{(1-n)\int p(x)\,dx}}, \end{aligned}$$

where C is a constant of integration. If $n = 1$, then equation (◇) becomes

$$\begin{aligned} \frac{dy}{dx} &= y\,(q-p) \\ \frac{dy}{y} &= (q-p)\,dx \\ y &= C_2\,e^{\int [q(x)-p(x)]\,dx}. \end{aligned}$$

The general solution is then, with C_1 and C_2 constants,

$$y = \begin{cases} \left[\dfrac{(1-n)\int e^{(1-n)\int p(x)\,dx}\,q(x)\,dx + C_1}{e^{(1-n)\int p(x)\,dx}}\right]^{1/(1-n)} & \text{for } n \neq 1 \\ C_2\,e^{\int [q(x)-p(x)]\,dx} & \text{for } n = 1. \end{cases}$$

REFERENCES:

Boyce, W. E. and DiPrima, R. C. *Elementary Differential Equations and Boundary Value Problems, 5th ed.* New York: Wiley, p. 28, 1992.

Ince, E. L. *Ordinary Differential Equations.* New York: Dover, p. 22, 1956.

Rainville, E. D. and Bedient, P. E. *Elementary Differential Equations.* New York: Macmillian, pp. 69-71, 1964.

Simmons, G. F. *Differential Equations, With Applications and Historical Notes.* New York: McGraw-Hill, p. 49, 1972.

Zwillinger, D. (Ed.). *CRC Standard Mathematical Tables and Formulae.* Boca Raton, FL: CRC Press, p. 413, 1995.

Zwillinger, D. "Bernoulli Equation." §II.A.37 in *Handbook of Differential Equations, 3rd ed.* Boston, MA: Academic Press, pp. 120 and 157-158, 1997.

la solución nos la da de forma implícita y hay que despejar la *g(x)*

```
> SolEcDif2g := solve(SolEcDif2,g(x));
```

$$SolEcDif2g := \frac{\left(2\sqrt{x} + \_C1\right)^2}{x} \quad \textbf{(10)}$$

Para graficarla debemos especificar la solución particular, vale decir el valor de las constantes. Lo hacemos de una forma arbitraria

```
> GrafSolEcDif2 := {seq(subs(_C1 = m, SolEcDif2g), m=-4..4)};
```

$$GrafSolEcDif2 := \left\{4, \frac{\left(2\sqrt{x}-4\right)^2}{x}, \frac{\left(2\sqrt{x}-3\right)^2}{x}, \frac{\left(2\sqrt{x}-2\right)^2}{x}, \frac{\left(2\sqrt{x}-1\right)^2}{x}, \frac{\left(2\sqrt{x}+1\right)^2}{x}, \frac{\left(2\sqrt{x}+2\right)^2}{x}, \frac{\left(2\sqrt{x}+3\right)^2}{x}, \frac{\left(2\sqrt{x}+4\right)^2}{x}\right\} \quad \textbf{(11)}$$

```
> plot(GrafSolEcDif2, x=4..10, 0..20);
```



¿Si, en vez de haber definido las constantes arbitrariamente hubieramos especificado las condiciones iniciales ?

o si lo hubiéramos resuelto por el método de las isoclinas

```
> DEplot(EcDif2,g(x),x=4..10, g= 0..10);
```



Nótese que para la región $0 < y < 4$ con $0 < x < \infty$ la solución presenta patologías
No así para la región $4 < y < \infty$ con $0 < x < \infty$

Probemos con ecuaciones Exactas. Ecuaciones de la forma $M(x, y) + N(x, y) \frac{d}{dx} y(x) = 0$

Vale decir $d(\Phi(x,y)) = 0 \Leftrightarrow d(\Phi(x,y)) = \frac{\partial}{\partial x} \Phi\, dx + \frac{\partial}{\partial y} \Phi\, dy = 0$ entonces

$\frac{\partial}{\partial y} M(x, y) = \frac{\partial}{\partial x} N(x, y)$

veamos

```
> restart;with(DEtools):infolevel[dsolve]:=5;
```

$$infolevel_{dsolve} := 5 \qquad \textbf{(12)}$$

```
> EcDif := x*(2*x^2 + y(x)^2) + y(x)*(x^2 + 2*y(x)^2)*diff(y(x),x)=0;
```

$$EcDif := x\,(2\,x^2 + y(x)^2) + y(x)\,(x^2 + 2\,y(x)^2)\left(\frac{d}{dx}\,y(x)\right) = 0 \qquad \textbf{(13)}$$

Antes de integrarla hay forma de tener idea del tipo de ecuación diferencial

```
> odeadvisor(EcDif);
```

(14)

$$[[\_homogeneous, class\ A], \_exact, \_rational, \_dAlembert] \quad (14)$$

o directamente

```
> DEtools[odeadvisor](EcDif);
```

$$[[\_homogeneous, class\ A], \_exact, \_rational, \_dAlembert] \quad (15)$$

De esta manera podemos conocer si MAPLE la identifica y podrá integrarla. Cada una de las características que MAPLE identifica puede ser buscado en la hoja de ayudas de MAPLE

Entonces procedem

```
> dsolve(EcDif,y(x));
Methods for first order ODEs:
--- Trying classification methods ---
trying a quadrature
trying 1st order linear
trying Bernoulli
trying separable
trying inverse linear
trying homogeneous types:
trying homogeneous D
<- homogeneous successful
```

$$y(x) = -\frac{1}{2}\frac{\sqrt{-2x^2\_C1 - 2\sqrt{-3\_C1^2x^4+4}}}{\sqrt{\_C1}}, y(x) = \frac{1}{2}\frac{\sqrt{-2x^2\_C1 - 2\sqrt{-3\_C1^2x^4+4}}}{\sqrt{\_C1}}, y(x) = -\frac{1}{2}\frac{\sqrt{-2x^2\_C1 + 2\sqrt{-3\_C1^2x^4+4}}}{\sqrt{\_C1}}, y(x) = \frac{1}{2}\frac{\sqrt{-2x^2\_C1 + 2\sqrt{-3\_C1^2x^4+4}}}{\sqrt{\_C1}} \quad (16)$$

La clasificación dice que es exacta, verifiquemos si cumple con las condiciones de exacta

```
> M := x*(2*x^2 + y^2); diffMdy := diff(M,y);
```

$$M := x\left(2x^2+y^2\right)$$
$$diffMdy := 2\,x\,y \quad (17)$$

```
> N :=  y*(x^2 + 2*y^2); diffNdx := diff(N,x);
```

$$N := y\left(x^2+2y^2\right)$$
$$diffNdx := 2\,x\,y \quad (18)$$

efectivamente $\frac{\partial}{\partial y}M(x,y) = \frac{\partial}{\partial x}N(x,y)$

consideremos otra ecuación diferencial

```
> EcDif := x*y(x)*ln(y(x)) + (x^2 + sqrt(y(x)^2 +1)*y(x)^2)*diff(y
  (x),x)=0;
```

(19)

$$EcDif := x\,y(x)\,\ln(y(x)) + \left(x^2 + \sqrt{y(x)^2+1}\;y(x)^2\right)\left(\frac{d}{dx}\,y(x)\right) = 0 \tag{19}$$

```
> DEtools[odeadvisor](EcDif);
 -> Computing symmetries using: way = HINT
   -> Calling odsolve with the ODE diff(y(x) x)+y(x)/x y(x)
   Methods for first order ODEs:
   --- Trying classification methods ---
   trying a quadrature
   trying 1st order linear
   <- 1st order linear successful
   -> Calling odsolve with the ODE diff(y(x) x)-y(x)/x y(x)
   Methods for first order ODEs:
   --- Trying classification methods ---
   trying a quadrature
   trying 1st order linear
   <- 1st order linear successful
-> The symmetry found is [1/(ln(y)^2*x) 0]
```

[[_1st_order, _with_symmetry_[F(x)*G(y),0]]]   **(20)**

Las simetrías que menciona MAPLE indican que no es exacta, pero un factor integrador la convierte en exacta. Esto es

Vale decir d(μ(x,y)Φ(x,y)) = 0 ⇔ d(μ(x,y)Φ(x,y)) =
$\mu(x,y)\,\frac{\partial}{\partial x}\,\Phi\,dx + \mu(x,y)\,\frac{\partial}{\partial y}\,\Phi\,dy = 0$ entonces $\frac{\partial}{\partial y}\left(\mu(x,y)M(x,y)\right) = \frac{\partial}{\partial x}\left(\mu(x,y)N(x,y)\right)$

y se puede averiguar cual es el factor integrador

```
> mu := intfactor(EcDif);
```

$$\mu := \frac{\ln(y(x))}{y(x)} \tag{21}$$

de esta forma

```
> DEtools[odeadvisor](mu*EcDif);
```

[_exact, [_1st_order, _with_symmetry_[F(x)*G(y),0]], _dAlembert]   **(22)**

se identifica como una ecuación diferenciale exacta

por fin la resolvemos

```
> dsolve(EcDif,y(x));
Methods for first order ODEs:
--- Trying classification methods ---
trying a quadrature
trying 1st order linear
trying Bernoulli
trying separable
trying inverse linear
trying homogeneous types:
trying Chini
differential order: 1; looking for linear symmetries
trying exact
<- exact successful
```

$$\frac{1}{2}\ln(y(x))^2 x^2 - \frac{1}{9}\sqrt{y(x)^2+1}\;y(x)^2 + \frac{1}{3}\,y(x)^2\ln(y(x))\sqrt{y(x)^2+1} + \frac{4}{9} \tag{23}$$

$$- \frac{4}{9}\sqrt{y(x)^2+1} - \frac{1}{3}\ln(y(x)) + \frac{1}{3}\ln(y(x))\sqrt{y(x)^2+1} + \frac{1}{3}\ln\Bigg(\frac{1}{2} + \frac{1}{2}\sqrt{y(x)^2+1}\Bigg) + \_C1 = 0$$

Otra ecuación diferencial

```
> EcDif := (3*x +2*y(x) +y(x)^2) + (x+4*x*y(x) +5*y(x)^2)*diff(y(x),
  x)=0;
```

$$EcDif := 3x + 2y(x) + y(x)^2 + \left(x + 4xy(x) + 5y(x)^2\right)\left(\frac{d}{dx}y(x)\right) = 0 \qquad \textbf{(24)}$$

```
> DEtools[odeadvisor](EcDif);
```

$$[\_rational] \qquad \textbf{(25)}$$

```
> dsolve(EcDif,y(x));
Methods for first order ODEs:
--- Trying classification methods ---
trying a quadrature
trying 1st order linear
trying Bernoulli
trying separable
trying inverse linear
trying homogeneous types:
trying Chini
differential order: 1; looking for linear symmetries
trying exact
Looking for potential symmetries
trying inverse_Riccati
trying an equivalence to an Abel ODE
differential order: 1; trying a linearization to 2nd order
--- trying a change of variables {x -> y(x), y(x) -> x}
differential order: 1; trying a linearization to 2nd order
trying 1st order ODE linearizable_by_differentiation
--- Trying Lie symmetry methods, 1st order ---
 -> Computing symmetries using: way = 2
```

$$\left[0, \frac{xy^2 + y^3 + x^2 + xy}{4xy + 5y^2 + x}\right]$$

```
 <- successful computation of symmetries.
trying an integrating factor from the invariance group
<- integrating factor successful
```

$$y(x) = -\frac{x\,RootOf\left(-1 + \_Z^5\,\_C1 + \left(-x^2 - x\right)\_Z^4 + 2x\,\_Z^2\right)^2 - 1}{RootOf\left(-1 + \_Z^5\,\_C1 + \left(-x^2 - x\right)\_Z^4 + 2x\,\_Z^2\right)^2} \qquad \textbf{(26)}$$

No siempre MAPLE tiene resultados. Miremos este ejemplo

```
> restart; infolevel[dsolve]:=2;
```

$$infolevel_{dsolve} := 2 \qquad \textbf{(27)}$$

```
> EcDif3 := diff(y(x),x)= (2*x -y(x)^2*sin(x*y(x)))/(cos(y(x)*x) -y
  (x)*x*sin(y(x)*x));
```

$$EcDif3 := \frac{d}{dx}\, y(x) = \frac{2\,x - y(x)^2 \sin(x\,y(x))}{\cos(x\,y(x)) - y(x)\,x \sin(x\,y(x))} \quad \textbf{(28)}$$

```
> EcDif31 := diff(y(x),x)*(cos(y(x)*x) -y(x)*x*sin(y(x)*x))= (2*x -y
  (x)^2*sin(x*y(x)));
```

$$EcDif31 := \left(\frac{d}{dx}\, y(x)\right) (\cos(x\,y(x)) - y(x)\,x \sin(x\,y(x))) = 2\,x - y(x)^2 \sin(x\,y(x)) \quad \textbf{(29)}$$

```
> DEtools[odeadvisor](EcDif3);DEtools[odeadvisor](EcDif31);
```

$$[y=\_G(x,y')]$$

$$[y=\_G(x,y')] \quad \textbf{(30)}$$

Que es una no lineal de la forma (
$2\,x - y^2 \sin(x\,y(x))\big)\, dx - (\cos(x\,y(x)) - y(x)\,x \sin(x\,y(x)))\, dy = 0$

Ecuación no lineal separable

```
> dsolve(EcDif3,y(x));
Methods for first order ODEs:
--- Trying classification methods ---
trying a quadrature
trying 1st order linear
trying Bernoulli
trying separable
trying inverse linear
trying homogeneous types:
trying Chini
differential order: 1; looking for linear symmetries
trying exact
Looking for potential symmetries
trying inverse_Riccati
trying an equivalence to an Abel ODE
differential order: 1; trying a linearization to 2nd order
--- trying a change of variables {x -> y(x), y(x) -> x}
differential order: 1; trying a linearization to 2nd order
trying 1st order ODE linearizable_by_differentiation
--- Trying Lie symmetry methods, 1st order ---
 -> Computing symmetries using: way = 3
 -> Computing symmetries using: way = 4
 -> Computing symmetries using: way = 5
trying symmetry patterns for 1st order ODEs
-> trying a symmetry pattern of the form [F(x)*G(y), 0]
-> trying a symmetry pattern of the form [0, F(x)*G(y)]
-> trying symmetry patterns of the forms [F(x),G(y)] and [G(y),F
(x)]
-> trying a symmetry pattern of the form [F(x),G(x)]
-> trying a symmetry pattern of the form [F(y),G(y)]
-> trying a symmetry pattern of the form [F(x)+G(y), 0]
-> trying a symmetry pattern of the form [0, F(x)+G(y)]
-> trying a symmetry pattern of the form [F(x),G(x)*y+H(x)]
-> trying a symmetry pattern of conformal type
```

Probemos si es exacta

```
> Coefdx := (2*x -y^2*sin(x*y)); diffyCoefdx := diff(Coefdx,y);
```

$$Coefdx := 2\,x - y^2 \sin(x\,y)$$

$$diffyCoefdx := -2\,y \sin(x\,y) - y^2 \cos(x\,y)\,x \tag{31}$$

```
> Coefdy := (cos(y*x) -y*x*sin(y*x)); diffxCoefdy := diff(Coefdy,x);
```

$$Coefdy := \cos(x\,y) - y\,x \sin(x\,y)$$

$$diffxCoefdy := -2\,y \sin(x\,y) - y^2 \cos(x\,y)\,x \tag{32}$$

bingo ! La ecuación es una ecuación diferencial exacta. Vale decir d(Φ(x,y)) =

$$\frac{\partial}{\partial x}\,\Phi\,dx + \frac{\partial}{\partial y}\,\Phi\,dy = 0$$

Con lo cual

```
> Phi := int(Coefdx,x) +f(y);
```

$$\Phi := x^2 + \cos(x\,y)\,y + f(y) \tag{33}$$

```
> EcDif12 := diff(Phi,y) = Coefdy;
```

$$EcDif12 := \cos(x\,y) - y\,x \sin(x\,y) + \frac{d}{dy}\,f(y) = \cos(x\,y) - y\,x \sin(x\,y) \tag{34}$$

```
> simplify(EcDif12);
```

$$\cos(x\,y) - y\,x \sin(x\,y) + \frac{d}{dy}\,f(y) = \cos(x\,y) - y\,x \sin(x\,y) \tag{35}$$

finalmente $\frac{d}{dy}\,f(y) = 0$ con lo cual $f(y) = Constante$

```
> Phi := x^2+cos(x*y(x))*y(x)+C;
```

$$\Phi := x^2 + \cos(x\,y(x))\,y(x) + C \tag{36}$$

y la solución queda de forma implícita

```
> diff(Phi,x);
```

$$2\,x - \sin(x\,y(x))\left(y(x) + x\left(\frac{d}{dx}\,y(x)\right)\right)y(x) + \left(\frac{d}{dx}\,y(x)\right)\cos(x\,y(x)) \tag{37}$$

```
> EcDif3;
```

$$\frac{d}{dx}\,y(x) = \frac{2\,x - y(x)^2 \sin(x\,y(x))}{\cos(x\,y(x)) - y(x)\,x \sin(x\,y(x))} \tag{38}$$